Stream
Model Independent Transaction
User Guide

User Guide for Release 2021.08

By

Jim Lewis

SynthWorks VHDL Training

Jim@SynthWorks.com

http://www.SynthWorks.com

Table of Contents

1. Overview

The Stream Model Independent Transaction package (StreamTransactionPkg.vhd) defines a record for communication and transaction initiation procedures that are suitable for Stream Interfaces.

All verification components (VCs) that use this interface, support a common set of transactions (or API) that are defined in StreamTransactionPkg.vhd. Hence, a user who understand the transactions for one verification component will also understand the transactions of another verification component that also uses the same package. This makes the job of a verification engineer easier.

The main difference between verification components that support this package is the configuration of verification component interface specific features - such as error injection for a UART or configuration of TID, TDest, TUser, and TLast for AxiStream.

Having a shared set of transactions improves test case reuse between different verification components – a case where reuse is rarely possible with other verification methodologies.

2. Stream Transaction Interface (StreamRecType)

The Stream Transaction Interface, StreamRecType shown in Figure 1, defines the transaction interface between the test sequencer and the verification component. As such, it is the primary channel for information exchange between the two.

```
type StreamRecType is record
  -- Handshaking controls
  --   Driven by RequestTransaction in the Transaction Procedures
  Rdy             : bit_max ;
  --   Driven by WaitForTransaction in the Verification Component
  Ack             : bit_max ;
  -- Transaction Type
  Operation       : StreamOperationType ;
  -- Data and Transaction Parameter to and from verification component
  DataToModel     : std_logic_vector_max_c ;
  ParamToModel    : std_logic_vector_max_c ;
  DataFromModel   : std_logic_vector_max_c ;
  ParamFromModel  : std_logic_vector_max_c ;
  -- BurstFifo
  BurstFifo       : ScoreboardIdType ;
  -- Verification Component Options Parameters - used by SetModelOptions
  IntToModel      : integer_max ;
  BoolToModel     : boolean_max ;
  IntFromModel    : integer_max ;
  BoolFromModel   : boolean_max ;
  TimeToModel     : time_max ;
  TimeFromModel   : time_max ;
  -- Verification Component Options Type
  Options         : integer_max ;
end record StreamRecType ;
```

Figure 1. StreamRecType

The record element types, bit_max, std_logic_vector_max_c, integer_max, time_max, and boolean_max, are defined in the OSVVM package ResolutionPkg. These types allow the record to support multiple drivers and use resolution functions based on function maximum (return largest value).

3. StreamOperationType

StreamOperationType is an enumerated type that indicates to the verification component type of transaction that is being dispatched. Being an enumerated type, it allows the determination of the operation in the simulator's waveform window. Table 2 shows the correlation between StreamOperationType values and the transaction name.

AddressBusOperationType Value	Transmitter Transaction Name	Receiver Transaction Name
WAIT_FOR_CLOCK	WaitForClock	WaitForClock
WAIT_FOR_TRANSACTION	WaitForTransaction	WaitForTransaction
GET_TRANSACTION_COUNT	GetTransactionCount	GetTransactionCount
GET_ALERTLOG_ID	GetAlertLogID	GetAlertLogID
SET_BURST_MODE	SetBurstMode	SetBurstMode
GET_BURST_MODE	GetBurstMode	GetBurstMode
SET_MODEL_OPTIONS	SetModelOptions	SetModelOptions
GET_MODEL_OPTIONS	GetModelOptions	GetModelOptions
SEND	Send	
SEND_ASYNC	SendAsync	
SEND_BURST	SendBurst	
SEND_BURST_ASYNC	SendBurstAsync	
GET		Get
TRY_GET		TryGet
GET_BURST		GetBurst
TRY_GET_BURST		TryGetBurst
CHECK		Check
TRY_CHECK		TryCheck
CHECK_BURST		CheckBurst
TRY_CHECK_BURST		TryCheckBurst

Figure 2. Correlation between StreamOperationType and the transaction name

4. Usage of the Transaction Interface (StreamRecType)

The data and parameter fields of the StreamRecType are unconstrained. Unconstrained objects may be used on component/entity interfaces. The record fields need to be sized by the record signal that is mapped as the actual in the test harness of the testbench. Figure 3 shows the declaration StreamTxRec (which connects the AxiStreamTransmitter to TestCtrl) and StreamRxRec (which connects the AxiStreamReceiver to TestCtrl).

```
constant AXI_PARAM_WIDTH : integer :=
      TID'length + TDest'length + TUser'length + 1;

signal StreamTxRec, StreamRxRec : StreamRecType(
    DataToModel   (AXI_DATA_WIDTH-1  downto 0),
    ParamToModel  (AXI_PARAM_WIDTH-1 downto 0),
    DataFromModel (AXI_DATA_WIDTH-1  downto 0),
    ParamFromModel(AXI_PARAM_WIDTH-1 downto 0)
  ) ;
```

Figure 3. StreamRecType

5. Types of Transactions

A transaction may be either a directive or an interface transaction.

Directive transactions interact with the verification component without generating any transactions or interface waveforms. Examples of these are WaitForClock and GetAlertLogID.

An interface transaction results in interface signaling to the DUT. An interface transaction may be either blocking (such as Send or Get) or non-blocking (such as SendAsync and TryGet).

A blocking transaction is an interface transaction that does not does not return (complete) until the interface operation requested by the transaction has completed.

An asynchronous transaction is nonblocking interface transaction that returns before the transaction has completed - typically immediately and before the transaction has started. An asynchronous transaction has "Async" as part of its name.

A Try transaction is nonblocking interface transaction that checks to see if transaction information is available, such as read data, and if it is returns it. A Try transaction has "Try" as part of its name.

6. Directive Transactions

Directive transactions interact with the verification component without generating any transactions or interface waveforms. These transactions are supported by all verification components.

```
------------------------------------------------------------
procedure WaitForTransaction (
--  Wait until pending (transmit) or next (receive) transaction(s) complete
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType
) ;

------------------------------------------------------------
```

```
procedure WaitForClock (
-- Wait for NumberOfClocks number of clocks
-- relative to the verification component clock
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  WaitCycles      : in    natural := 1
) ;

------------------------------------------------------------
procedure GetTransactionCount (
-- Get the number of transactions handled by the model.
------------------------------------------------------------
  signal    TransactionRec   : inout StreamRecType ;
  variable  TransactionCount : out   integer
) ;

------------------------------------------------------------
procedure GetAlertLogID (
-- Get the AlertLogID from the verification component.
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  variable  AlertLogID      : out   AlertLogIDType
) ;

------------------------------------------------------------
procedure GetErrorCount (
-- Error reporting for testbenches that do not use OSVVM AlertLogPkg
-- Returns error count.  If an error count /= 0, also print errors
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  variable  ErrorCount      : out   natural
) ;
```

7. Burst FIFOs and Burst Mode Controls

The burst FIFOs hold bursts of data that is to be sent to or received from the interface. The burst FIFO can be configured in the modes defined for StreamFifoBurstModeType. Currently these modes defined as a subtype of integer, shown below. The intention is to facilitate model specific extensions without the need to define separate transactions.

```
subtype StreamFifoBurstModeType is integer ;

-- Word mode indicates the burst FIFO contains interface words.
-- The size of the word may either be interface specific (such as
-- a UART which supports up to 8 bits) or be interface instance specific
-- (such as AxiStream which supports interfaces sizes of 1, 2, 4, 8,
-- 16, ... bytes)
constant STREAM_BURST_WORD_MODE       : StreamFifoBurstModeType := 0 ;
```

```
-- Word + Param mode indicates the burst FIFO contains interface
-- words plus a parameter.   The size of the parameter is also either
-- interface specific (such as the OSVVM UART, which uses 3 bits -
-- one bit for each of parity, stop, and break error injection) or
-- interface instance specific (such as AxiStream which uses the Param
-- field to hold TUser).  AxiStream TUser may be different size for
-- different applications.
constant STREAM_BURST_WORD_PARAM_MODE : StreamFifoBurstModeType  := 1 ;

-- Byte mode is experimental and may be removed in a future revision.
-- Byte mode indicates that the burst FIFO contains bytes.
-- The verification component assembles interface words from the bytes.
-- This allows transfers to be conceptualized in an interface independent
-- manner.
constant STREAM_BURST_BYTE_MODE       : StreamFifoBurstModeType := 2 ;
```

SetBurstMode and GetBurstMode are directive transactions that configure the burst mode into one of the modes defined in for StreamFfifoBurstModeType.

```
------------------------------------------------------------
procedure SetBurstMode (
------------------------------------------------------------
  signal   TransRec    : InOut StreamRecType ;
  constant OptVal      : In    StreamFifoBurstModeType
) ;

------------------------------------------------------------
procedure GetBurstMode (
------------------------------------------------------------
  signal   TransRec    : InOut StreamRecType ;
  variable OptVal      : Out   StreamFifoBurstModeType
) ;
```

8. Set and Get Model Options

Model operations are directive transactions that are used to configure the verification component. They can either be used directly or with a model specific wrapper around them - see AXI VCs for an example.

```
------------------------------------------------------------
procedure SetModelOptions (
------------------------------------------------------------
  signal   TransactionRec  : InOut StreamRecType ;
  constant Option          : In    integer ;
  constant OptVal          : In    boolean
) ;

------------------------------------------------------------
procedure SetModelOptions (
------------------------------------------------------------
  signal   TransactionRec  : InOut StreamRecType ;
```

```
    constant Option          : In    integer ;
    constant OptVal          : In    integer
  ) ;

  ------------------------------------------------------------
  procedure SetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer ;
    constant OptVal          : In    std_logic_vector
  ) ;

  ------------------------------------------------------------
  procedure SetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer ;
    constant OptVal          : In    time
  ) ;

  ------------------------------------------------------------
  procedure SetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer
  ) ;

  ------------------------------------------------------------
  procedure GetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer ;
    variable OptVal          : Out   boolean
  ) ;

  ------------------------------------------------------------
  procedure GetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer ;
    variable OptVal          : Out   integer
  ) ;

  ------------------------------------------------------------
  procedure GetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer ;
```



```
    variable OptVal          : Out   std_logic_vector
  ) ;

  ------------------------------------------------------------
  procedure GetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer ;
    variable OptVal          : Out   time
  ) ;

  ------------------------------------------------------------
  procedure GetModelOptions (
  ------------------------------------------------------------
    signal   TransactionRec  : InOut StreamRecType ;
    constant Option          : In    integer
  ) ;
```

9. Transmitter Transactions

9.1 Send

Blocking Send Transaction. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for error injection.

```
  ------------------------------------------------------------
  procedure Send (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    constant  Data            : in    std_logic_vector ;
    constant  Param           : in    std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure Send (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    constant  Data            : in    std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

9.2 SendAsync

SendAsync is an asynchronous / non-blocking send transaction. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for error injection.

```
  ------------------------------------------------------------
  procedure SendAsync (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
```

```
    constant  Data            : in    std_logic_vector ;
    constant  Param           : in    std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure SendAsync (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    constant  Data            : in    std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

9.3 SendBurst

SendBurst is a blocking send burst transaction. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for error injection.

```
  ------------------------------------------------------------
  procedure SendBurst (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    constant  NumFifoWords    : In    integer ;
    constant  Param           : in    std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE   ) ;

  ------------------------------------------------------------
  procedure SendBurst (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    constant  NumFifoWords    : In    integer ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

9.4 SendBurstAsync

SendBurstAsync is an asynchronous / non-blocking send burst transaction. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for error injection.

```
  ------------------------------------------------------------
  procedure SendBurstAsync (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    constant  NumFifoWords    : In    integer ;
    constant  Param           : in    std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure SendBurstAsync (
  ------------------------------------------------------------
```

```
    signal    TransactionRec  : inout StreamRecType ;
    constant  NumFifoWords    : In    integer ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

10.Receiver Transactions

10.1 Get

Get is a blocking get transaction. Param, when present, is an extra parameter used by the verification component.

```
  ------------------------------------------------------------
  procedure Get (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  Data            : out   std_logic_vector ;
    variable  Param           : out   std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure Get (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  Data            : out   std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

10.2 TryGet

TryGet is a non-blocking try get transaction. If Data is available, get it and return available TRUE, otherwise Return Available FALSE. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for received error status.

```
  ------------------------------------------------------------
  procedure TryGet (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  Data            : out   std_logic_vector ;
    variable  Available       : out   boolean ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure TryGet (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  Data            : out   std_logic_vector ;
    variable  Param           : out   std_logic_vector ;
```

```
    variable  Available       : out   boolean ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

10.3 GetBurst

GetBurst is a blocking get burst transaction. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for checking error injection.

```
  ------------------------------------------------------------
  procedure GetBurst (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  NumFifoWords    : inout integer ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure GetBurst (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  NumFifoWords    : inout integer ;
    variable  Param           : out   std_logic_vector ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;
```

10.4 TryGetBurst

TryGetBurst is a non-blocking try get burst transaction. If Data is available, get it and return available TRUE, otherwise Return Available FALSE. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for received error status.

```
  ------------------------------------------------------------
  procedure TryGetBurst (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  NumFifoWords    : inout integer ;
    variable  Available       : out   boolean ;
    constant  StatusMsgOn     : in    boolean := FALSE
  ) ;

  ------------------------------------------------------------
  procedure TryGetBurst (
  ------------------------------------------------------------
    signal    TransactionRec  : inout StreamRecType ;
    variable  NumFifoWords    : inout integer ;
    variable  Param           : out   std_logic_vector ;
    variable  Available       : out   boolean ;
    constant  StatusMsgOn     : in    boolean := FALSE
```

```
) ;
```

10.5 Check

Check is a blocking check transaction. Data is the expected value to be received. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for received error status.

```
------------------------------------------------------------
procedure Check (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  Data            : in    std_logic_vector ;
  constant  Param           : in    std_logic_vector ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;

------------------------------------------------------------
procedure Check (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  Data            : in    std_logic_vector ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;
```

10.6 TryCheck

TryCheck is a non-blocking try check transaction If Data is available, check it and return available TRUE, otherwise Return Available FALSE. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for received error status.

```
------------------------------------------------------------
procedure TryCheck (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  Data            : in    std_logic_vector ;
  constant  Param           : in    std_logic_vector ;
  variable  Available       : out   boolean ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;

------------------------------------------------------------
procedure TryCheck (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  Data            : in    std_logic_vector ;
  variable  Available       : out   boolean ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;
```

10.7 CheckBurst

CheckBurst is a blocking check burst transaction. Param, when present, is an extra parameter used by the verification component. The UART verification component uses Param for checking error injection.

```
------------------------------------------------------------
procedure CheckBurst (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  NumFifoWords    : In    integer ;
  constant  Param           : in    std_logic_vector ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;

------------------------------------------------------------
procedure CheckBurst (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  NumFifoWords    : In    integer ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;
```

10.8 TryCheckBurst

TryCheckBurst is a non-blocking try check burst transaction. Param, when present, is an extra parameter used by the verification component. If BURST Data is available, check it and return available TRUE, otherwise Return Available FALSE. The UART verification component uses Param for checking error injection.

```
------------------------------------------------------------
procedure TryCheckBurst (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  NumFifoWords    : In    integer ;
  constant  Param           : in    std_logic_vector ;
  variable  Available       : out   boolean ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;

------------------------------------------------------------
procedure TryCheckBurst (
------------------------------------------------------------
  signal    TransactionRec  : inout StreamRecType ;
  constant  NumFifoWords    : In    integer ;
  variable  Available       : out   boolean ;
  constant  StatusMsgOn     : in    boolean := FALSE
) ;
```

11. Verification Component Support Functions

Verification component support functions help decode the operation value (StreamOperationType) to determine properties about the operation.

```
------------------------------------------------------------
function IsTry (
-- True when this transaction is an asynchronous or try transaction.
------------------------------------------------------------
  constant Operation     : in StreamOperationType
) return boolean ;

------------------------------------------------------------
function IsCheck (
-- True when this transaction is a check transaction.
------------------------------------------------------------
  constant Operation     : in StreamOperationType
) return boolean ;
```

12. Burst FIFOs Initiator

12.1 BurstFifo is in the Interface

The BurstFifo is inside StreamRecType, see Figure 4. This makes the BurstFifo easily accessible to both the Verification component as well as the Test Sequencer (TestCtrl). The BurstFifo is implemented using a ScoreboardID from the scoreboard package. This allows a VC to either use it as a FIFO or as a scoreboard. The FIFO is std_logic_vector based and uses the OSVVM library ScoreboardGenericPkg instance defined in ScoreboardPkg_slv.vhd (OsvvmLibraries/osvvm).

```
type StreamRecType is record
  . . .
  -- BurstFifo
  BurstFifo       : ScoreboardIdType ;
  . . .
end record StreamRecType ;
```

Figure 4. BurstFifo In StreamRecType

12.2 Using the Burst FIFOs in the VC

12.2.1 Initializing Burst FIFOs

The burst FIFOs need to be initialized. A good place to do this is in the transaction dispatcher of the verification components. Figure 5 shows the declaration of a BurstFifo.

```
TransactionDispatcher : process
 . . .
begin
  wait for 0 ns ;
  TransRec.BurstFifo <= NewID("RxBurstFifo", ModelID) ;
  wait for 0 ns ;
```

Figure 5. BurstFifo Initialization

12.2.2 Accessing Burst FIFOs

The Burst Fifos support basic FIFO operations. These are shown in Figure 6.

```
Push(TransRec.BurstFifo, Data) ;
Check(TransRec.BurstFifo, Data) ;
Data := Pop(TransRec.BurstFifo) ;
```

Figure 6. Making the BurstFifos visible in the test sequencer (TestCtrl)

12.2.3 Packing and Unpacking the FIFO

The burst FIFOs can be configured to be either byte width or match the verification component interface width.   The following procedures (from FifoFillPkg_slv.vhd) are used to transform byte width data in the burst FIFO to/from the verification component interface width.

```
------------------------------------------------------------
procedure PopWord (
-- Pop bytes from BurstFifo and form a word
-- Current implementation for now assumes it is assembling bytes.
--
------------------------------------------------------------
  constant Fifo              : in    ScoreboardIDType ;
  variable Valid             : out   boolean ;
  variable Data              : out   std_logic_vector ;
  variable BytesToSend       : inout integer ;
  constant ByteAddress       : in    natural := 0
) ;

------------------------------------------------------------
procedure PushWord (
-- Push a word into the byte oriented BurstFifo
-- Current implementation for now assumes it is assembling bytes.
--
------------------------------------------------------------
  constant Fifo              : in    ScoreboardIDType ;
  variable Data              : in    std_logic_vector ;
  constant DropUndriven      : in    boolean := FALSE ;
  constant ByteAddress       : in    natural := 0
) ;

------------------------------------------------------------
procedure CheckWord (
-- Check a word using the byte oriented BurstFifo
-- Current implementation for now assumes it is assembling bytes.
--
------------------------------------------------------------
  constant Fifo              : in    ScoreboardIDType ;
```

```
    variable Data              : in    std_logic_vector ;
    constant DropUndriven      : in    boolean := FALSE ;
    constant ByteAddress       : in    natural := 0
  ) ;
```

12.3 Using the Burst FIFO in the Test Sequencer

12.3.1 Filling the Burst FIFO from the Test Sequencer

In the test sequencer, to send a burst to the transmitter or check a burst in the transmitter, the following procedures from FifoFillPkg_slv.vhd (in osvvm_common library) may be used.

```
  ------------------------------------------------------------
  procedure PushBurst (
  -- Push each value in the VectorOfWords parameter into the FIFO.
  -- Only FifoWidth bits of each value will be pushed.
  ------------------------------------------------------------
    constant Fifo           : in    ScoreboardIDType ;
    constant VectorOfWords  : in    integer_vector ;
    constant FifoWidth      : in    integer := 8
  ) ;

  ------------------------------------------------------------
  procedure PushBurstIncrement (
  -- Push Count number of values into FIFO.  The first value
  -- pushed will be FirstWord and following values are one greater
  -- than the previous one.
  -- Only FifoWidth bits of each value will be pushed.
  ------------------------------------------------------------
    constant Fifo      : in    ScoreboardIDType ;
    constant FirstWord : in    integer ;
    constant Count     : in    integer ;
    constant FifoWidth : in    integer := 8
  ) ;

  ------------------------------------------------------------
  procedure PushBurstRandom (
  -- Push Count number of values into FIFO.  The first value
  -- pushed will be FirstWord and following values are randomly generated
  -- using the first value as the randomization seed.
  -- Only FifoWidth bits of each value will be pushed.
  ------------------------------------------------------------
    constant Fifo      : in    ScoreboardIDType ;
    constant FirstWord : in    integer ;
    constant Count     : in    integer ;
    constant FifoWidth : in    integer := 8
  ) ;
```

12.3.2 Reading and/or Checking the Read Burst in the Test Sequencer

The following PopBurst and CheckBurst are used in the test sequencer to verify received burst values.

```
------------------------------------------------------------
procedure PopBurst (
-- Pop values from the FIFO into the VectorOfWords parameter.
-- Each value popped will be FifoWidth bits wide.
------------------------------------------------------------
  constant Fifo           : in    ScoreboardIDType ;
  variable VectorOfWords  : out   integer_vector ;
  constant FifoWidth      : in    integer := 8
) ;

------------------------------------------------------------
procedure CheckBurst (
-- Pop values from the FIFO and check them against each value
-- in the VectorOfWords parameter.
-- Each value popped will be FifoWidth bits wide.
------------------------------------------------------------
  constant Fifo           : in    ScoreboardIDType ;
  constant VectorOfWords  : in    integer_vector ;
  constant FifoWidth      : in    integer := 8
) ;

------------------------------------------------------------
procedure CheckBurstIncrement (
-- Pop values from the FIFO and check them against values determined
-- by an incrementing pattern.  The first check value will be FirstWord
-- and the following check values are one greater than the previous one.
-- Each value popped will be FifoWidth bits wide.
------------------------------------------------------------
  constant Fifo      : in    ScoreboardIDType ;
  constant FirstWord : in    integer ;
  constant Count     : in    integer ;
  constant FifoWidth : in    integer := 8
) ;

------------------------------------------------------------
procedure CheckBurstRandom (
-- Pop values from the FIFO and check them against values determined
-- by a random pattern.  The first check value will be FirstWord and the
-- following check values are randomly generated using the first
-- value as the randomization seed.
-- Each value popped will be FifoWidth bits wide.
------------------------------------------------------------
  constant Fifo      : in    ScoreboardIDType ;
  constant FirstWord : in    integer ;
  constant Count     : in    integer ;
```

```
    constant FifoWidth : in    integer := 8
  ) ;
```

12.3.3 Example: Sending Bursts via the Transmitter

The following are transactions initiated by the AxiStreamTransmitter verification component (see TbStream_SendGetBurst1.vhd).

```
constant DATA_WIDTH : integer := 32 ;
. . .

AxiTransmitterProc : process
begin
  . . .
  log("Transmit 32 Bytes -- word aligned") ;
  PushBurstIncrement(TxRec.BurstFifo, 3, 32, DATA_WIDTH) ;
  SendBurst(TxRec, 32) ;

  WaitForClock(TxRec, 4) ;

  log("Transmit 30 Bytes -- unaligned") ;
  PushBurst(TxRec.BurstFifo, (1,3,5,7,9,11,13,15,17,19,21,23), WIDTH) ;
  PushBurst(TxRec.BurstFifo, (31,33,35,37,39,41,43,45,47,49,1,3), WIDTH) ;
  SendBurst(TxRec, 30) ;

  WaitForClock(TxRec, 4) ;

  log("Transmit 34 Bytes -- unaligned") ;
  PushBurstRandom(TxRec.BurstFifo, 7, 34, DATA_WIDTH) ;
  SendBurst(TxRec, 34) ;
```

12.3.4 Example: Getting Bursts via the Receiver

The following are transactions initiated by the AxiStreamReceiver verification component (see TbStream_SendGetBurst1.vhd).

```
AxiReceiverProc : process
  variable NumBytes : integer ;
begin
  WaitForClock(RxRec, 2) ;

--    log("Transmit 32 Bytes -- word aligned") ;
  GetBurst (RxRec, NumBytes) ;
  AffirmIfEqual(NumBytes, 32, "Receiver: NumBytes Received") ;
  CheckBurstIncrement(RxRec.BurstFifo, 3, NumBytes, DATA_WIDTH) ;

--    log("Transmit 30 Bytes -- unaligned") ;
  GetBurst (RxRec, NumBytes) ;
  AffirmIfEqual(NumBytes, 30, "Receiver: NumBytes Received") ;
  CheckBurst(RxRec.BurstFifo, (1,3,5,7,9,11,13,15,17,19,21,23), WIDTH);
```

```
  CheckBurst(RxRec.BurstFifo, (31,33,35,37,39,41,43,45,47,49,1,3), WIDTH);

--    log("Transmit 34 Bytes -- unaligned") ;
  GetBurst (RxRec, NumBytes) ;
  AffirmIfEqual(NumBytes, 34, "Receiver: NumBytes Received") ;
  CheckBurstRandom(RxRec.BurstFifo, 7, NumBytes, DATA_WIDTH) ;
```

12.3.5 Example: Checking Bursts in the Receiver

The same bursts that were read in the receiver can also be checked in the receiver (see TbStream_SendCheckBurst1.vhd).

```
AxiReceiverProc : process
  variable NumBytes : integer ;
begin
  WaitForClock(RxRec, 2) ;

--    log("Transmit 32 Bytes -- word aligned") ;
  PushBurstIncrement(RxRec.BurstFifo, 3, 32, FIFO_WIDTH) ;
  CheckBurst(RxRec, 32) ;

  WaitForClock(RxRec, 4) ;

--    log("Transmit 30 Bytes -- unaligned") ;
  PushBurst(RxRec.BurstFifo, (1,3,5,7,9,11,13,15,17,19,21,23), WIDTH);
  PushBurst(RxRec.BurstFifo, (31,33,35,37,39,41,43,45,47,49,1,3), WIDTH);
  CheckBurst(RxRec, 30) ;

  WaitForClock(RxRec, 4) ;

--    log("Transmit 34 Bytes -- unaligned") ;
  PushBurstRandom(RxRec.BurstFifo, 7, 34, FIFO_WIDTH) ;
  CheckBurst(RxRec, 34) ;
```

13. About the OSVVM Model Independent Transactions

OSVVM Model Independent Transactions were developed and are maintained by Jim Lewis of SynthWorks VHDL Training. These evolved from methodology and packages developed for SynthWorks' VHDL Testbenches and verification class. They are part of the Open Source VHDL Verification Methodology (OSVVM) model library (osvvm_common), which brings leading edge verification techniques to the VHDL community.

Please support OSVVM by purchasing your VHDL training from SynthWorks.

14. About the Author - Jim Lewis

Jim Lewis, the founder of SynthWorks, has thirty plus years of design, teaching, and problem solving experience. In addition to working as a Principal Trainer for SynthWorks, Mr Lewis has done ASIC and FPGA design, custom model development, and consulting.

Mr. Lewis is chair of the IEEE 1076 VHDL Working Group (VASG) and is the primary developer of the Open Source VHDL Verification Methodology (OSVVM.org) packages. Neither of these activities generate revenue. Please support our volunteer efforts by buying your VHDL training from SynthWorks.

If you find bugs these packages or would like to request enhancements, you can reach me at jim@synthworks.com.

15. References

[1] Jim Lewis, VHDL Testbenches and Verification, student manual for SynthWorks' class.